

5. Expressões aritméticas

DIM0320

2015.1

Sumário

- 1 Operadores inteiros
- 2 Precedências
- 3 Alteração de prioridades
- 4 Operadores reais
- 5 Expressões de tipo misto
- 6 Exercícios

- 1 Operadores inteiros
- 2 Precedências
- 3 Alteração de prioridades
- 4 Operadores reais
- 5 Expressões de tipo misto
- 6 Exercícios

Operadores inteiros

Operadores aritméticos

Operações que podem ser executadas sobre números.

Lista

Suponha que x seja uma variável de tipo inteiro.

Operador	Exemplos
$+$, $-$	-3 , $+x$
$\backslash$	$5 \backslash 2$
$+$, $-$, $*$	$4 + 8 * 9$
mod ou $\%$	$3 \% 2$

Definição (Resto/divisão inteira.)

$\forall a, b \in \mathbb{Z}^2, \exists q, r \in \mathbb{N}^2,$

$$a = q * b + r, |r| < |b|$$

Resto

$$r = a \% b$$

Divisão inteira/euclidiana

$$q = a \backslash b$$

Expressões

Expressões inteiras válidas

- Cada operador atua sobre elementos do **mesmo tipo**.
 - ▶ É preciso variáveis e constantes do mesmo tipo (*inteiro*).

Tipo dos operadores vistos:

Binários ($\backslash$, $\%$, $+$, $-$, $*$) $\mathbb{Z} \times \mathbb{Z} \rightarrow \mathbb{Z}$

Unários ($+$, $-$) $\mathbb{Z} \rightarrow \mathbb{Z}$

Dúvida ?

- Seja r : real. Tipo de $r + 2$?
 - ▶ real ? inteiro?
 - ▶ ter uma única resposta = aplicar regras semânticas

- 1 Operadores inteiros
- 2 Precedências**
- 3 Alteração de prioridades
- 4 Operadores reais
- 5 Expressões de tipo misto
- 6 Exercícios

Problema

Valor de $5 * 3 \% 2$?

- **Regras de precedências** são usadas para desambiguar tais expressões, i.e. para poder escrevê-las sem parênteses.

Prioridades

Níveis de prioridade (prec)

Conjuntos de operadores entre (..) definem **classes de equivalência**.

$$(+, -, \text{unário}) \succ (*, \backslash, \%) \succ (+, -, \text{binário})$$

Associatividade à esquerda

- Sejam p, q dois operadores em conflitos
 - $p \succ q$, a operação p é executada antes de q .
 - $q \prec p$, a operação q é executada antes de p .
 - $p \approx q$, a operação **mais à esquerda** é executada antes da outra.

Exemplo (Precedências)

- $a + b + c \rightarrow (a + b) + c$
- $a + b * c \% 2 \rightarrow (a + ((b * c) \% 2))$

Exemplo

```
algoritmo "Expressoes aritmeticas envolvendo variaveis e constantes"
var
    a , b , c : inteiro
inicio
    escreva("Entre com o valor da variavel a: ")
    leia(a)

    escreva("Entre com o valor da variavel b: ")
    leia(b)

    escreva("Entre com o valor da variavel c: ")
    leia(c)

    escreval("O resultado da expressao a * b % c e: ", a * b % c)

    escreval("O resultado da expressao -a * b % c * 8 e: ", -a * b % c * 8 )
    escreval("O resultado da expressao -a - b - c * 3 e: ", -a - b - c * 3 )
fimalgoritmo
```

- 1 Operadores inteiros
- 2 Precedências
- 3 Alteração de prioridades**
- 4 Operadores reais
- 5 Expressões de tipo misto
- 6 Exercícios

Uso de parênteses

Pode-se alterar as prioridades com (...) (como em matemática).

Exemplo (Ilustração)

① $(a + b) * (c \% 2)$

Isto é $op_1 * op_2$ com:

▶ $op_1 = a + b$

▶ $op_2 = c \% 2$

Execeturamos + , depois %, enfim *.

② $((2 + 3) - (1 + 2)) * 3 - (3 + (3 - 2)) ??$

Instrução / comando de atribuição

Usando expressões (inteiras), precisaremos armazenar o valor de algumas delas, através comandos de atribuição (<- ou leia).

- leia poder ser usada apenas quando a variável é uma variável de **entrada**.
- <- é usado dentro do corpo do algoritmo para armazenar valores, intermediários, ou seja, o resultado.

Exemplo de atribuições

```
algoritmo "Quadrado da soma de 2 inteiros"
var a , b , quadrado : inteiro
inicio
  escreva("Entre com o primeiro inteiro: ")
  leia(a)

  escreva("Entre com o segundo inteiro: ")
  leia(b)

  quadrado <- (a + b) * (a + b)

  escreva ("O quadrado da soma dos inteiros lidos e: " , quadrado)
fimalgoritmo
```

- 1 Operadores inteiros
- 2 Precedências
- 3 Alteração de prioridades
- 4 Operadores reais**
- 5 Expressões de tipo misto
- 6 Exercícios

Operadores reais

Sintaxe

Um constante do tipo `real` é

- uma sequência de números terminada por `.`
- uma sequência de números começada por `.`
- uma sequência de números, um `.`, um outra sequência de números

Lista

Suponha `x : real`

Operador	Exemplos
<code>+</code> , <code>-</code>	<code>-3.2</code> , <code>+x</code>
<code>+</code> , <code>-</code> , <code>*</code> , <code>/</code>	<code>4. + 8.3 * 9.3</code>
<code>^</code>	<code>3.3 ^ 2.1</code>

Precedências

Níveis de prioridade

Conjuntos de operadores entre (..) definem **classes de equivalência**.

$$\{+, -(unário)\} \succ \{^, *, /\} \succ \{+, -(binário)\}$$

Associatividade à esquerda

- Sejam p, q dois operadores em conflitos
 - $p \succ q$, a operação p é executada antes de q .
 - $q \prec p$, a operação q é executada antes de p .
 - $p \approx q$, a operação **mais à esquerda** é executada antes da outra.

Exemplo (Precedências)

- $a - b + c \rightarrow (a - b) + c$
- $a + b * c / 2 \rightarrow (a + ((b * c) / 2))$

Uso de parênteses

Expressão a converter

$$(3.0 * a + 2.0 * b) / (c - (a - 1.0) / (1.0 + (a + b) / (2.0 * c)))$$

```
algoritmo "Expressoes aritmeticas com variaveis e constantes reais"
var a, b, c, res : real
inicio
  escreva("Entre com o valor da variavel a: ")
  leia(a)

  escreva("Entre com o valor da variavel b: ")
  leia(b)

  escreva("Entre com o valor da variavel c: ")
  leia(c)

  res <- (3.0 * a + 2.0 * b) / c - (a - 1.0) / (1.0 + (a + b) / (2.0 * c))
  escreva ("O resultado da expressao e: ", res)
finalgoritmo
```

Tipos dos operadores

Resumo

Operador	Tipos
$+, -$	$\mathbb{Z} \times \mathbb{Z} \rightarrow \mathbb{Z}, \mathbb{Z} \rightarrow \mathbb{Z}, \mathbb{Z} \times \mathbb{R} \rightarrow \mathbb{R}, \mathbb{R} \rightarrow \mathbb{R}$
$\backslash, \%$	$\mathbb{Z} \times \mathbb{Z} \rightarrow \mathbb{Z}$
$*$	$\mathbb{Z} \times \mathbb{Z} \rightarrow \mathbb{Z}, \mathbb{R} \times \mathbb{R} \rightarrow \mathbb{R}$
$/, ^$	$\mathbb{R} \times \mathbb{R} \rightarrow \mathbb{R}$

Observação

Em Portugal:

- $\mathbb{Z} \approx$ inteiro
- $\mathbb{R} \approx$ real

- 1 Operadores inteiros
- 2 Precedências
- 3 Alteração de prioridades
- 4 Operadores reais
- 5 Expressões de tipo misto**
- 6 Exercícios

Misturar tipos

- Nos exemplos aritméticos usamos só variáveis e constantes do tipo real ou variáveis e constantes do tipo inteiro.

Misturar tipos

- Nos exemplos aritméticos usamos só variáveis e constantes do tipo real ou variáveis e constantes do tipo inteiro.
- Em matemática, $3, 14 + 2 = 5, 14$ porque $\mathbb{Z} \subset \mathbb{R}$

Misturar tipos

- Nos exemplos aritméticos usamos só variáveis e constantes do tipo real ou variáveis e constantes do tipo inteiro.
- Em matemática, $3, 14 + 2 = 5, 14$ porque $\mathbb{Z} \subset \mathbb{R}$
- Podemos fazer isso em Portugol?

Regras semânticas

Regra

Em expressões de tipo misto:

- ① o tipo do operando mais simples é convertido para o tipo do operando mais complexo.
- ② a operação aritmética é aplicada a operandos de mesmo tipo.

Operador de conversão $\uparrow$

Para **marcar** as etapas de conversões, usaremos o operador seguinte

$$\uparrow: \mathbb{Z} \rightarrow \mathbb{R}$$

Exemplos

Exemplo (Etapas de conversão)

a, b: real, c: inteiro

$$\begin{aligned}(a - b) * (c/2) &\rightarrow (a - b) * (c/2 \uparrow) \\ &\rightarrow (a - b) * (c \uparrow / 2 \uparrow)\end{aligned}$$

Exemplo (Etapas de conversão)

a, b, c: inteiro

$$\begin{aligned}(a - b) * (c/2) &\rightarrow (a - b) * (c/2 \uparrow) \\ &\rightarrow (a - b) * (c \uparrow / 2 \uparrow) \\ &\rightarrow (a - b) \uparrow * (c \uparrow / 2 \uparrow)\end{aligned}$$

Uso da divisão real

Divisão real

As regras semânticas permitem o uso da divisão real $/$ e de exponenciação $^$ com tipos inteiros.

Exemplo (Validade de expressões)

- $5 / 2$
- $5 \% 2 / 3$
- $5 / 2 \% 2$

Observação

Nenhum valor real pode ser automaticamente convertido em um valor inteiro.

Precedências mistas

Níveis de prioridade

Conjuntos de operadores entre (..) definem **classes de equivalência**.

$$\{+, -(unário)\} \succ \{^, \backslash, \%, *, /\} \succ \{+, -(binário)\}$$

Associatividade à esquerda

- Sejam p, q dois operadores em conflitos
 - $p \succ q$, a operação p é executada antes de q .
 - $p \prec q$, a operação q é executada antes de p .
 - $p \approx q$, a operação **mais à esquerda** é executada antes da outra.

Exemplo (Precedências)

- $a - b + c \rightarrow (a - b) + c$
- $a + b / c \% 2 \rightarrow (a + ((b / c) \% 2))$

Algoritmo com valores reais e inteiros

Assunto

Desenvolver um algoritmo para determinar o volume, V , de uma esfera a partir do raio, r , da esfera.

$$\text{Formula: } V = r^3 * \pi * 4 / 3$$

Observação

π é uma constante, palavra reservada da linguagem Portugol.

Solução

```
algoritmo "Volume da esfera"
var raio : real
    volume : real
inicio
    escreva ("Entre com o valor do raio da esfera: ")
    leia (raio)

    volume <- (4 / 3) * pi * raio * raio * raio

    escreva ("O volume da esfera de raio ", raio, " e ", volume)
fimalgoritmo
```

Algumas primitivas

Exponencição

`exp(a, b) = a ^ b`

Valores aleatórios

`rand()` retorna um número real gerado aleatoriamente, maior ou igual a zero e menor que um.

Truncamento

`int(e)`

- **Exemplos**

- ▶ `int(3.14) = 3`

- ▶ `int(-3.5) = -3`

Representação dos tipos aritméticos

Inteiros

A representação dos tipos inteiros é exata $\mathbb{Z}$

Reais

- A representação do tipo real não é exata.
- Alguns elementos de $\mathbb{R}$ não podem ser representados no tipo real (ex: $1/3$).
- Os cálculos reais podem produzir erros de precisão (ver ponto flutuante e perda de precisão).

Conclusão

É melhor usar inteiros até precisar da conversão.

Resumo

- Expressões aritméticas reais
- Operadores reais
- Expressões aritméticas mistas
- Primitivas
- Uso dos tipos aritméticos

Perguntas ?



<http://dimap.ufrn.br/~richard/dim0320>

- 1 Operadores inteiros
- 2 Precedências
- 3 Alteração de prioridades
- 4 Operadores reais
- 5 Expressões de tipo misto
- 6 Exercícios**

Primeiros exercícios

- 1 Escrever em Portugol $5x^3 + 7x^2 - 3x - 1$
- 2 Avaliar, de acordo com as regras de precedências:

▶ $2 - 3 * 5$

▶ $1 \setminus 4 - 8 * 4 * 10 \setminus 9$

▶ $3 * 5 + 1 - 6 \% 2 \setminus 3$

- 3 Suponha que

- 1 a linha 11 seja substituída por

```
1 quadrado <- a + b * a + b
```

O algoritmo continuará correto ?

- 2 a linha 11 seja substituída por as duas seguintes linhas

```
1 quadrado <- a + b
```

```
2 quadrado <- quadrado * quadrado
```

O algoritmo continuará correto ?

Somas de cubos

Assunto

Escrever um algoritmo calculando a soma dos cubos de 2 números inteiros lidos como entradas. Essa soma será a saída (escrita) do algoritmo.

Somas de cubos

Assunto

Escrever um algoritmo calculando a soma dos cubos de 2 números inteiros lidos como entradas. Essa soma será a saída (escrita) do algoritmo.

Solução

```
algoritmo "Cubos"
var m, n : inteiro
inicio
    escreval("Entre com um inteiro: ")
    leia(m)
    escreval("Entre com um inteiro: ")
    leia(n)
    escreval("A soma dos cubos e: ", m * m * m + n * n * n)
fimalgoritmo
```

Tradução

Traduzir em Português a notação

$$\frac{1}{1 + \frac{1}{1 + a}}$$

- ❶ se a é um valor real;
- ❷ se a é um valor inteiro

Área de retângulo

Assunto

Escrever um algoritmo que leia os valores correspondentes à base e altura de um retângulo, determine a área do retângulo e escreva a área como saída.

Área de retângulo

Assunto

Escrever um algoritmo que leia os valores correspondentes à base e altura de um retângulo, determine a área do retângulo e escreva a área como saída.

Solução

```
algoritmo "Area retangulo"
var altura, base: real
inicio
  escreva("Entre com valores positivo para a base do retangulo: ")
  leia(base)
  escreva("Entre com valores positivo para a altura do retangulo: ")
  leia(altura)
  escreval("A area do retangulo e:", altura * base)
fimalgoritmo
```

Área de círculo

Assunto

- Escrever um algoritmo para calcular a área de um círculo.
- A entrada do seu algoritmo é o valor do raio do círculo.
- A saída é o valor da área do círculo.

Área de círculo

Assunto

- Escrever um algoritmo para calcular a área de um círculo.
- A entrada do seu algoritmo é o valor do raio do círculo.
- A saída é o valor da área do círculo.

Solução

```
algoritmo "Area circulo"
var raio: real
inicio
  escreva("Entre com um valor positivo para o raio do circulo: ")
  leia(raio)
  escreval("A area do circulo e:", pi * raio * raio)
fimalgoritmo
```

Valores aleatórios de um intervalo

Assunto

Use a primitiva `rand()` para escrever um algoritmo que gera um valor aleatório do intervalo $[a, b]$, onde a e b são dois valores dados.